
AshPhoenix
Welcome! This is the package for integrating Phoenix Framework and Ash Framework. It provides tools for integrating with Phoenix forms (AshPhoenix.Form), Phoenix LiveViews (AshPhoenix.LiveView) and more.

  
    
  
  Installation


Add ash_phoenix to your list of dependencies in mix.exs:
{:ash_phoenix, "~> 2.1.8"}

  
    
  
  Whats in the box?


	AshPhoenix.Form - A form data structure for using resource actions with phoenix forms
	AshPhoenix.Form.Auto - Tools to automatically determine nested form structures based on calls to manage_relationship for an action.
	AshPhoenix.FilterForm - A form data structure for building filter statements
	AshPhoenix.LiveView - Helpers for querying data and integrating changes
	AshPhoenix.SubdomainPlug - A plug to determine a tenant using subdomains for multitenancy
	AshPhoenix.FormData.Error - A protocol to allow errors to be rendered in forms
	Phoenix.HTML.Safe implementations for Ash.CiString, Ash.NotLoaded and Decimal
	mix ash_phoenix.gen.live for generating liveview modules
	mix ash_phoenix.gen.html for generating controllers and views


  
    
  
  Tutorials


	Getting Started with Ash and Phoenix


  
    
  
  Topics


	Union Forms



  

    
Get Started with Ash and Phoenix
    


  
    
  
  Goals


In this guide we will:
	Create a new Phoenix project
	Setup Ash, AshPhoenix and AshPostgres as dependencies
	Create a basic Blog.Post resource
	Create and migrate the database
	Learn how to interact with your resource
	Integrate a minimal Phoenix LiveView with Ash


  
    
  
  Preparation


	Install Elixir
	Phoenix - Up and Running Guide
	Design Principles


  
    
  
  Requirements


If you want to follow along yourself, you will need the following things:
	Elixir (1.12 or later) and Erlang (22 or later) installed
	PostgreSQL installed
	A text editor
	A terminal to run the examples


  
    
  
  Setup



  
    
  
  Create a New Phoenix Project



  
    
  
  Install Phoenix


This section is based on the Phoenix installation docs. For more details go there.

First we need to install the Phoenix project generator, then we'll run the generator to create our new project.
# install Phoenix project generator
$ mix archive.install hex phx_new

# generate Phoenix project
$ mix phx.new my_ash_phoenix_app

# cd into project
$ cd my_ash_phoenix_app


  
    
  
  Don't run mix ecto.create


Do not run mix ecto.create, (as it asks you to) we will do this the Ash way later.


  
    
  
  Add Dependencies


We now need to add Ash, AshPhoenix and AshPostgres to our Phoenix project. We need to add the dependencies to the deps function in our mix.exs. We'll also need to add dependencies to our .formatter.exs to ensure consistent formatting when using mix format.
# mix.exs

  def deps do
    [
      # use `mix hex.info <library_name>` to get the latest versions of each dependency, for example, `mix hex.info phoenix`
      {:phoenix, "~> x.x"},
      # ...
      {:ash, "~> x.x"},
      {:picosat_elixir, "~> x.x"},
      {:ash_postgres, "~> x.x"},
      {:ash_phoenix, "~> x.x"}
    ]
  end
Add :ash, :ash_phoenix, and :ash_postgres to your .formatter.exs file.
# .formatter.exs
[
  import_deps: [..., :ash, :ash_phoenix, :ash_postgres],
  ...
]
Now in the terminal install these new dependencies.
$ mix deps.get


  
    
  
  Picosat installation issues?


If you have trouble compiling picosat_elixir, then replace {:picosat_elixir, "~> 0.2"} with {:simple_sat, "~> 0.1"} to use a simpler (but mildly slower) solver. You can always switch back to picosat_elixir later once you're done with the tutorial.


  
    
  
  Use AshPostgres.Repo


We need to swap Ecto.Repo for AshPostgres.Repo. AshPostgres.Repo enriches your repo with additional AshPostgres specific behaviour, but is essentially a thin wrapper around Ecto.Repo.
To use AshPostgres.Repo change your repo module to look like this:
# lib/my_ash_phoenix_app/repo.ex

defmodule MyAshPhoenixApp.Repo do
  use AshPostgres.Repo, otp_app: :my_ash_phoenix_app

  # Installs extensions that ash commonly uses
  def installed_extensions do
    ["ash-functions", "uuid-ossp", "citext"]
  end
end

  
    
  
  Edit Config


We need to specify the Ash domains that our application uses.
Add this to your config:
# config/config.exs

import Config

config :my_ash_phoenix_app,
  ash_domains: [MyAshPhoenixApp.Blog]

  
    
  
  Create the Domain and add Resources


An Ash domain can be thought of as a Bounded Context in Domain Driven Design terms and can seen as analogous to a Phoenix context. Put simply, its a way of grouping related resources together. In our case our domain will be called MyAshPhoenixApp.Blog.
An Ash domain points to Ash resources. An Ash domain can point to one or more resources. In our case we will only have a single resource MyAshPhoenixApp.Blog.Post. We'll be taking a deeper look into that in the next section.
For now take a look at the Blog domain and the associated resources:
# lib/my_ash_phoenix_app/blog/blog.ex

defmodule MyAshPhoenixApp.Blog do
  use Ash.Domain

  resources do
    resource MyAshPhoenixApp.Blog.Post do
      # Define an interface for calling resource actions.
      define :create_post, action: :create
      define :list_posts, action: :read
      define :update_post, action: :update
      define :destroy_post, action: :destroy
      define :get_post, args: [:id], action: :by_id
    end
  end
end

  
    
  
  Creating Resources



  
    
  
  Creating the Post Resource


A resource is a central concept in Ash. In short, a resource is a domain model object in your system. A resource defines the data it holds and defines the actions that can operate on that data.
When we create Post we will place it in lib/my_ash_phoenix_app/blog/post.ex. So the structure after making the resource should look like so:
lib/
├─ my_ash_phoenix_app/
│  ├─ blog/
│  │  ├─ blog.ex
│  │  ├─ post.ex
Below is the resource module. Read the comments carefully, every line is explained:
# lib/my_ash_phoenix_app/blog/post.ex

defmodule MyAshPhoenixApp.Blog.Post do
  # Using Ash.Resource turns this module into an Ash resource.
  use Ash.Resource,
    # Tells Ash where the generated code interface belongs
    domain: MyAshPhoenixApp.Blog,
    # Tells Ash you want this resource to store its data in Postgres.
    data_layer: AshPostgres.DataLayer

  # The Postgres keyword is specific to the AshPostgres module.
  postgres do
    # Tells Postgres what to call the table
    table "posts"
    # Tells Ash how to interface with the Postgres table
    repo MyAshPhoenixApp.Repo
  end

  actions do
    # Exposes default built in actions to manage the resource
    defaults [:read, :destroy]

    create :create do
      # accept title as input
      accept [:title]
    end

    update :update do
      # accept content as input
      accept [:content]
    end

    # Defines custom read action which fetches post by id.
    read :by_id do
      # This action has one argument :id of type :uuid
      argument :id, :uuid, allow_nil?: false
      # Tells us we expect this action to return a single result
      get? true
      # Filters the `:id` given in the argument
      # against the `id` of each element in the resource
      filter expr(id == ^arg(:id))
    end
  end

  # Attributes are simple pieces of data that exist in your resource
  attributes do
    # Add an autogenerated UUID primary key called `:id`.
    uuid_primary_key :id
    # Add a string type attribute called `:title`
    attribute :title, :string do
      # We don't want the title to ever be `nil`
      allow_nil? false
    end

    # Add a string type attribute called `:content`
    # If allow_nil? is not specified, then content can be nil
    attribute :content, :string
  end
end

  
    
  
  Creating and Migrating the Database


We have specified the resource in Ash. But we have yet to create it in our data layer (in our case Postgres).
First we need to create our database:
$ mix ash.setup

Running setup for AshPostgres.DataLayer...
The database for MyAshPhoenixApp.Repo has been created

01:23:45.678 [info] Migrations already up

Now we need to populate our database. We do this by generating and performing a migration.
We can use a generator to produce a migration for us. Ash can deduce what needs to go into the migration and do the hard work for us, to do this use the command below:
$ mix ash.codegen initial_migration

# ... don't worry about other files it creates

Generating Migrations:
* creating priv/repo/migrations/20230208045101_initial_migration.exs

Here is the migration file commented in detail:
# priv/repo/migrations/20230208045101_initial_migration.exs

defmodule MyAshPhoenixApp.Repo.Migrations.InitialMigration do
  use Ecto.Migration

  # This function runs when migrating forward
  def up do
    # Creates the `:posts` table
    create table(:posts, primary_key: false) do
      # Adds primary key attribute `:id` of type `:uuid`
      # null values are not allowed
      add :id, :uuid, null: false, default: fragment("gen_random_uuid()"), primary_key: true

      # Adds attribute `:title` of type `:text`, null values are not allowed
      add :title, :text, null: false
      # Adds attribute `:content` of type `:text`, null values are allowed
      add :content, :text
    end
  end

  # This is the function that runs if you want to rollback the migration.
  def down do
    # Deletes the `:posts` table
    drop table(:posts)
  end
end
We can run the up/0 function which will perform the desired operations on the Postgres database. We do this with the migrate command:
$ mix ash.migrate

In case you want to drop the database and start over again during development you can use mix ash.reset.


  
    
  
  Interacting with your Resources


All interaction with your resource attributes always occur through an action. In our resource we are using the default actions for :create, :read, :update, :destroy along with a custom action :by_id.
:create and :update and :destroy actions require a changeset. Ash changesets are conceptually similar to Ecto changesets. They're data structures which represent an intended change to an Ash resource and provide validation.
The :read action takes a query instead of a changeset.
Below is the most verbose way of calling your resource. All other ways of interaction are some kind of shorthand of these. This means at some point a changeset is being created and passed to the domain, even if it's encapsulated within another function.
# create post
new_post =
  MyAshPhoenixApp.Blog.Post
  |> Ash.Changeset.for_create(:create, %{title: "hello world"})
  |> Ash.create!()

# read all posts
MyAshPhoenixApp.Blog.Post
|> Ash.Query.for_read(:read)
|> Ash.read!()

# get single post by id
MyAshPhoenixApp.Blog.Post
|> Ash.Query.for_read(:by_id, %{id: new_post.id})
|> Ash.read_one!()

# update post
updated_post =
  new_post
  |> Ash.Changeset.for_update(:update, %{content: "hello to you too!"})
  |> Ash.update!()

# delete post
new_post
|> Ash.Changeset.for_destroy(:destroy)
|> Ash.destroy!()
As stated above, this is verbose so Ash has a built in shortcut - The code_interface. You may notice this has already been done in your Post resource inside of the domain module.

  
    
  
  you can call code interfaces whatever you like


The function name doesn't have to match the action name in any way. You could also write:
define :make_post, action: :create
That's perfectly valid and could be called via Blog.make_post/2.

Now we can call our resource like so:
# create post
new_post = MyAshPhoenixApp.Blog.create_post!(%{title: "hello world"})

# read post
MyAshPhoenixApp.Blog.list_posts!()

# get post by id
MyAshPhoenixApp.Blog.get_post!(new_post.id)

# update post
updated_post = MyAshPhoenixApp.Blog.update_post!(new_post, %{content: "hello to you too!"})

# delete post
MyAshPhoenixApp.Blog.destroy_post!(updated_post)
Now isn't that more convenient?

  
    
  
  raising and non-raising functions


All functions that interact with an Ash resource have a raising and non-raising version. For example there are two create functions create/2 and create!/2. create/2 returns {:ok, resource} or {:error, reason}. create!/2 will return just the record on success and will raise an error on failure.


  
    
  
  Connecting your Resource to a Phoenix LiveView


Now we know how to interact with our resource, let's connect it to a simple Phoenix LiveView. Here is the LiveView below. If you are using phoenix live_view <= 0.18, you will need to use let={} instead of :let={}.
# lib/my_ash_phoenix_app_web/posts_live.ex

defmodule MyAshPhoenixAppWeb.PostsLive do
  use MyAshPhoenixAppWeb, :live_view

  alias MyAshPhoenixApp.Blog
  alias MyAshPhoenixApp.Blog.Post

  def render(assigns) do
    ~H"""
    <h2 class="text-xl text-center">Your Posts</h2>
    <div class="my-4">
      <div :if={Enum.empty?(@posts)} class="font-bold text-center">
        No posts created yet
      </div>
      <ol class="list-decimal">
        <li :for={post <- @posts} class="mt-4">
          <div class="font-bold"><%= post.title %></div>
          <div><%= if Map.get(post, :content), do: post.content, else: "" %></div>
          <button
            class="mt-2 p-2 bg-black text-white rounded-md"
            phx-click="delete_post"
            phx-value-post-id={post.id}
          >
            Delete post
          </button>
        </li>
      </ol>
    </div>
    <h2 class="mt-8 text-lg">Create Post</h2>
    <.form :let={f} for={@create_form} phx-submit="create_post">
      <.input type="text" field={f[:title]} placeholder="input title" />
      <.button class="mt-2" type="submit">Create</.button>
    </.form>
    <h2 class="mt-8 text-lg">Update Post</h2>
    <.form :let={f} for={@update_form} phx-submit="update_post">
      <.label>Post Name</.label>
      <.input type="select" field={f[:post_id]} options={@post_selector} />
      <.input type="text" field={f[:content]} placeholder="input content" />
      <.button class="mt-2" type="submit">Update</.button>
    </.form>
    """
  end

  def mount(_params, _session, socket) do
    posts = Blog.list_posts!()

    socket =
      assign(socket,
        posts: posts,
        post_selector: post_selector(posts),
        create_form: AshPhoenix.Form.for_create(Post, :create) |> to_form(),
        update_form: AshPhoenix.Form.for_update(List.first(posts, %Post{}), :update) |> to_form()
      )

    {:ok, socket}
  end

  def handle_event("delete_post", %{"post-id" => post_id}, socket) do
    post_id |> Blog.get_post!() |> Blog.destroy_post!()
    posts = Blog.list_posts!()

    {:noreply, assign(socket, posts: posts, post_selector: post_selector(posts))}
  end

  def handle_event("create_post", %{"form" => form_params}, socket) do
    case AshPhoenix.Form.submit(socket.assigns.create_form, params: form_params) do
      {:ok, _post} ->
        posts = Blog.list_posts!()

        {:noreply, assign(socket, posts: posts, post_selector: post_selector(posts))}

      {:error, create_form} ->
        {:noreply, assign(socket, create_form: create_form)}
    end
  end

  def handle_event("update_post", %{"form" => form_params}, socket) do
    case AshPhoenix.Form.submit(socket.assigns.update_form, params: form_params) do
      {:ok, _post} ->
        posts = Blog.list_posts!()

        {:noreply, assign(socket, posts: posts, post_selector: post_selector(posts))}

      {:error, update_form} ->
        {:noreply, assign(socket, update_form: update_form)}
    end
  end

  defp post_selector(posts) do
    for post <- posts do
      {post.title, post.id}
    end
  end
end
Don't forget to add the LiveView to your router.
# lib/my_ash_phoenix_app_web/router.ex
  scope "/", MyAshPhoenixAppWeb do
    # ...
    live "/posts", PostsLive
  end
Now, start the web server by running mix phx.server. Then, visit http://localhost:4000/posts in your browser to see what we have just created.
You can see how using functions created by our code_interface makes it easy to integrate Ash with Phoenix.
You may also notice this is the first time we've used the AshPhoenix library. The AshPhoenix library contains utilities to help Ash integrate with Phoenix and LiveView Seamlessly. One of these utilities is AshPhoenix.Form which can automatically produce changesets to be used in the forms.
That's it for this guide. We've gone from 0 to a fully working Phoenix App using Ash. To get a closer look, see the accompanying repo here.

  
    
  
  Where to Next?


We are really just scratching the surface of what can be done in Ash. Look below for what to look at next.

  
    
  
  Continue Learning


There's a few places you can go to learn more about how to use ash:
	Read more about how to query the data in your resources - Ash.Query
	Dig deeper into actions.
	Study resource relationship management


  
    
  
  Ash Authentication & Ash Authentication Phoenix


See the power Ash can bring to your web app or API. Get authentication working in minutes.

  
    
  
  Add an API (or two)


Check out the AshJsonApi and AshGraphql extensions to effortlessly build APIs around your resources.


  

    
Union Forms
    

When building a form for a union, you use inputs_for as normal, but a few things are done for you under the hood.
Lets take this example union:
defmodule NormalContent do
  use Ash.Resource, data_layer: :embedded

  attributes do
    attribute :body, :string, allow_nil?: false
  end

  actions do
    defaults [:read, create: [:body], update: [:body]]
  end
end

defmodule SpecialContent do
  use Ash.Resource, data_layer: :embedded

  attributes do
    attribute :text, :string, allow_nil?: false
  end

  actions do
    defaults [:read, create: [:text], update: [:text]]
  end
end

defmodule Content do
  use Ash.Type.NewType,
    subtype_of: :union,
    constraints: [
      types: [
        normal: [
          type: NormalContent,
          tag: :type,
          tag_value: :normal
        ],
        special: [
          type: SpecialContent,
          tag: :type,
          tag_value: :special
        ]
      ]
    ]
end

  
    
  
  Determining the type for a union form


We track the type of the value in a hidden param called _union_type. You can use this to show a different form depending on the type of thing.

  
    
  
  Changing the type of a union form


If you want to let the user change the union type, you would use AshPhoenix.Form.remove_form/3 and AshPhoenix.Form.add_form/3. See the example below for the template, and here is an example event handler
def handle_event("type-changed", %{"_target" => path} = params, socket) do
  new_type = get_in(params, path)
  # The last part of the path in this case is the field name
  path = :lists.droplast(path)

  form =
    socket.assigns.form
    |> AshPhoenix.Form.remove_form(path)
    |> AshPhoenix.Form.add_form(path, params: %{"_union_type" => new_type})

  {:noreply, assign(socket, :form, form)}
end

  
    
  
  Non-embedded types


If one of your union values is a non embedded type, like :integer, it will still be a nested form, but you would access the single value with <.input field={nested_form[:value]} .../>

  
    
  
  Example


We might have a form like this:
<.inputs_for :let={fc} field={@form[:content]}>
  <!-- Dropdown for setting the union type -->
  <.input
    field={fc[:_union_type]}
    phx-change="type-changed"
    type="select"
    options={[Normal: "normal", Special: "special"]}
  />

  <!-- switch on the union type to display a form -->
  <%= case fc.params["_union_type"] do %>
    <% "normal" -> %>
      <.input  type="text" field={fc[:body]} />
    <% "special" -> %>
      <.input type="text" field={fc[:text]} />
  <% end %>
</.inputs_for>


  

    
Change Log
    

All notable changes to this project will be documented in this file.
See Conventional Commits for commit guidelines.

  
    
  
  v2.1.8 (2024-10-29)



  
    
  
  Improvements:


	track raw_params


  
    
  
  v2.1.7 (2024-10-29)



  
    
  
  Bug Fixes:


	set _union_type param when unnesting a resource in a union

	don't wrap resources inside of unions as WrappedValue

	warn on missing params on submit

	unwrap unions & wrapped values when fetching values



  
    
  
  v2.1.6 (2024-10-17)



  
    
  
  Improvements:


	allow phoenix_live_view rc


  
    
  
  v2.1.5 (2024-10-14)



  
    
  
  Improvements:


	support generic actions (#250)


  
    
  
  v2.1.4 (2024-09-30)



  
    
  
  Bug Fixes:


	properly include calc args in to_filter_map


  
    
  
  v2.1.3 (2024-09-30)



  
    
  
  Bug Fixes:


	properly apply calculations with arguments in filter form


  
    
  
  v2.1.2 (2024-09-03)



  
    
  
  Bug Fixes:


	spec update_form to accept functions of lists


  
    
  
  v2.1.1 (2024-08-01)



  
    
  
  Bug Fixes:


	Use :public? instead of :private? (#221)


  
    
  
  Improvements:


	raise an error on usage of old option name


  
    
  
  v2.1.0 (2024-07-26)



  
    
  
  Bug Fixes:


	ensure we prepare_source for all read action forms


  
    
  
  v2.0.4 (2024-06-13)



  
    
  
  Bug Fixes:


	various fixes for union form handling

	properly fill union values on update

	ensure changing union type is reflected in param transformer



  
    
  
  Improvements:


	honor _union_type when applying input


  
    
  
  v2.0.3 (2024-06-05)



  
    
  
  Bug Fixes:


	properly (i.e safely) encode ci strings to iodata and params

	various union param handling fixes

	properly transform nested params

	make sure that _index is correctly updated before and after removal for sparse forms (#196) (#197)



  
    
  
  v2.0.2 (2024-05-22)



  
    
  
  Bug Fixes:


	don't assume all embeds have a create/update action


  
    
  
  v2.0.1 (2024-05-17)



  
    
  
  Bug Fixes:


	improve union handling

	Convert entered action names into atoms for lookup in the resource (#187)

	various fixes around union forms



  
    
  
  Improvements:


	support adding a form by inserting to an index


  
    
  
  v2.0.0 (2024-04-30)


The changelog is being restarted. See /documentation/1.0-CHANGELOG.md in GitHub for the old changelog.

  
    
  
  Improvements:


	[AshPhoenix.Form] better error message with hints for accepted/non accepted missing forms


  
    
  
  Bug Fixes:


	[AshPhoenix.Form] don't use public_attributes?, check for all accepted attributes. In Ash 3.0, private attributes can be accepted

	[AshPhoenix.Form]

	Pass the tenant to Ash.can/3 and Ash.can?/3. (#165)

	Pass the tenant to Ash.can/3 and Ash.can?/3.




  

    
AshPhoenix.LiveView 
    



      
Utilities for keeping Ash query results up to date in a LiveView.

      


      
        Summary


  
    Types
  


    
      
        
  
    
    AshPhoenix.SubdomainPlug - ash_phoenix v2.1.8
    
    

    


  
  

    
AshPhoenix.SubdomainPlug 
    



      
This is a basic plug that loads the current tenant assign from a given
value set on subdomain.
This was copied from Triplex.SubdomainPlug, here:
  https://github.com/ateliware/triplex/blob/master/lib/triplex/plugs/subdomain_plug.ex
Options:
	:endpoint (atom/0) - Required. The endpoint that the plug is in, used for deterining the host

	:assign (atom/0) - The key to use when assigning the current tenant The default value is :current_tenant.

	:handle_subdomain - An mfa to call with the conn and a subdomain value. Can be used to do something like fetch the current user given the tenant. Must return the new conn.


To plug it on your router, you can use:
plug AshPhoenix.SubdomainPlug,
  endpoint: MyApp.Endpoint
An additional helper here can be used for determining the host in your liveview, and/or using
the host that was already assigned to the conn.
For example:
def handle_params(params, uri, socket) do
  socket =
    assign_new(socket, :current_tenant, fn ->
      AshPhoenix.SubdomainPlug.live_tenant(socket, uri)
    end)

  socket =
    assign_new(socket, :current_organization, fn ->
      if socket.assigns[:current_tenant] do
        MyApp.Accounts.Ash.get!(MyApp.Accounts.Organization,
          subdomain: socket.assigns[:current_tenant]
        )
      end
    end)

  {:noreply, socket}
end

      


      
        Summary


  
    Functions
  


    
      
        
  
    
    AshPhoenix.Form - ash_phoenix v2.1.8
    
    

    


  
  

    
AshPhoenix.Form 
    



      
A module to allow you to fluidly use resources with Phoenix forms.

  
    
  
  Life cycle


The general workflow is, with either LiveView or Phoenix forms:
	Create a form with AshPhoenix.Form
	Render the form with Phoenix.Component.form (or CoreComponents.simple_form), or, if using Surface, <Form>
	To validate the form (e.g with phx-change for liveview), pass the submitted params to AshPhoenix.Form.validate/3
	On form submission, pass the params to AshPhoenix.Form.submit/2
	On success, use the result to redirect or assign. On failure, reassign the provided form.

The following keys exist on the form to show where in the lifecycle you are:
	submitted_once? - If the form has ever been submitted. Useful for not showing any errors on the first attempt to fill out a form.
	just_submitted? - If the form has just been submitted and no validation has happened since. Useful for things like
triggering a UI effect that should stop when the form is modified again.
	.changed? - If something about the form is different than it originally was. Note that in some cases this can yield a
false positive, specifically if a nested form is removed and then a new one is added with the exact same values.
	.touched_forms - A MapSet containing all keys in the form that have been modified. When submitting a form, only these keys are included in the parameters.


  
    
  
  Working with related data


If your resource action accepts related data, (for example a managed relationship argument, or an embedded resource attribute), you can
use Phoenix's inputs_for for that field, but you must do one of two things:
	Tell AshPhoenix.Form to automatically derive this behavior from your action, for example:

form =
  user
  |> AshPhoenix.Form.for_update(:update, forms: [auto?: true])
  |> to_form()
	Explicitly configure the behavior of it using the forms option. See for_create/3 for more.

For example:
form =
  user
  |> AshPhoenix.Form.for_update(:update,
    forms: [
      profile: [
        resource: MyApp.Profile,
        data: user.profile,
        create_action: :create,
        update_action: :update
        forms: [
          emails: [
            data: user.profile.emails,
            resource: MyApp.UserEmail,
            create_action: :create,
            update_action: :update
          ]
        ]
      ]
    ])
  |> to_form()

  
    
  
  LiveView


AshPhoenix.Form (unlike ecto changeset based forms) expects to be reused throughout the lifecycle of the liveview.
You can use Phoenix events to add and remove form entries and submit/2 to submit the form, like so:
def render(assigns) do
  ~H"""
  <.simple_form for={@form} phx-change="validate" phx-submit="submit">
    <%!-- Attributes for the parent resource --%>
    <.input type="email" label="Email" field={@form[:email]} />
    <%!-- Render nested forms for related data --%>
    <.inputs_for :let={item_form} field={@form[:items]}>
      <.input type="text" label="Item" field={item_form[:name]} />
      <.input type="number" label="Amount" field={item_form[:amount]} />
      <.button type="button" phx-click="remove_form" phx-value-path={item_form.name}>
        Remove
      </.button>
    </.inputs_for>
    <:actions>
      <.button type="button" phx-click="add_form" phx-value-path={@form[:items].name}>
        Add Item
      </.button>
      <.button>Save</.button>
    </:actions>
  </.simple_form>
  """
end

def mount(_params, _session, socket) do
  form =
    MyApp.Grocery.Order
    |> AshPhoenix.Form.for_create(:create,
      forms: [
        items: [
          type: :list,
          resource: MyApp.Grocery.Item,
          create_action: :create
        ]
      ]
    )
    |> AshPhoenix.Form.add_form([:items])
    |> to_form()

  {:ok, assign(socket, form: form)}
end

# In order to use the `add_form` and `remove_form` helpers, you
# need to make sure that you are validating the form on change
def handle_event("validate", %{"form" => params}, socket) do
  form = AshPhoenix.Form.validate(socket.assigns.form, params)
  {:noreply, assign(socket, form: form)}
end

def handle_event("submit", %{"form" => params}, socket) do
  case AshPhoenix.Form.submit(socket.assigns.form, params: params) do
    {:ok, order} ->
      {:noreply,
       socket
       |> put_flash(:info, "Saved order for #{order.email}!")
       |> push_navigate(to: ~p"/")}

    {:error, form} ->
      {:noreply, assign(socket, form: form)}
  end
end

def handle_event("add_form", %{"path" => path}, socket) do
  form = AshPhoenix.Form.add_form(socket.assigns.form, path)
  {:noreply, assign(socket, form: form)}
end

def handle_event("remove_form", %{"path" => path}, socket) do
  form = AshPhoenix.Form.remove_form(socket.assigns.form, path)
  {:noreply, assign(socket, form: form)}
end

AshPhoenix.Form.Auto 
    



      
A tool to automatically generate available nested forms based on a resource and action.
To use this, specify forms: [auto?: true] when creating the form.
Keep in mind, you can always specify these manually when creating a form by simply specifying the forms option.
There are two things that this builds forms for:
	Attributes/arguments who's type is an embedded resource.
	Arguments that have a corresponding change manage_relationship(..) configured.

For more on relationships see the documentation for Ash.Changeset.manage_relationship/4.
When building forms, you can switch on the action type and/or resource of the form, in order to have different
fields depending on the form. For example, if you have a simple relationship called :comments with
on_match: :update and on_no_match: :create, there are two types of forms that can be in inputs_for(form, :comments).
In which case you may have something like this:
<%= for comment_form <- inputs_for(f, :comments) do %>
  <%= hidden_inputs_for(comment_form) %>
  <%= if comment_form.source.type == :create do %>
    <%= text_input comment_form, :text %>
    <%= text_input comment_form, :on_create_field %>
  <% else %>
    <%= text_input comment_form, :text %>
    <%= text_input comment_form, :on_update_field %>
  <% end %>

  <button phx-click="remove_form" phx-value-path="<%= comment_form.name %>">Add Comment</button>
  <button phx-click="add_form" phx-value-path="<%= comment_form.name %>">Add Comment</button>
<% end %>
This also applies to adding forms of different types manually. For instance, if you had a "search" field
to allow them to search for a record (e.g in a liveview), and you had an on_lookup read action, you could
render a search form for that read action, and once they've selected a record, you could render the fields
to update that record (in the case of on_lookup: :relate_and_update configurations).

  
    
  
  Options


	:relationship_fetcher (term/0) - A two argument function that receives the parent data, the relationship to fetch.
The default simply fetches the relationship value, and if it isn't loaded, it uses [] or nil.

	:sparse_lists? (boolean/0) - Sets all list type forms to sparse?: true by default. Has no effect on forms derived for embedded resources. The default value is false.

	:include_non_map_types? (boolean/0) - Creates form for non map or array of map type inputs The default value is false.



  
    
  
  Special Considerations



  
    
  
  on_lookup: :relate_and_update


For on_lookup: :relate_and_update configurations, the "read" form for that relationship will use the appropriate read action.
However, you may also want to include the relevant fields for the update that would subsequently occur. To that end, a special
nested form called :_update is created, that uses an empty instance of that resource as the base of its changeset. This may require
some manual manipulation of that data before rendering the relevant form because it assumes all the default values. To solve for this,
if you are using liveview, you could actually look up the record using the input from the read action, and then use AshPhoenix.Form.update_form/3
to set that looked up record as the data of the _update form.

  
    
  
  Many to Many Relationships


In the case that a manage_change option points to a join relationship, that form is presented via a special nested form called
_join. So the first form in inputs_for(form, :relationship) would be for the destination, and then inside of that you could say
inputs_for(nested_form, :_join). The parameters are merged together during submission.

AshPhoenix.Form.WrappedValue 
    



      
A sentinal value used when editing a union that has non-map values

AshPhoenix.FormData.Error protocol
    



      
A protocol for allowing errors to be rendered into a form.
To implement, define a to_form_error/1 and return a single error or list of errors of the following shape:
{:field_name, message, replacements}
Replacements is a keyword list to allow for translations, by extracting out the constants like numbers from the message.

AshPhoenix.FilterForm 
    



      
A module to help you create complex forms that generate Ash filters.
# Create a FilterForm
filter_form = AshPhoenix.FilterForm.new(MyApp.Payroll.Employee)
FilterForm's comprise 2 concepts, predicates and groups. Predicates are the simple boolean
expressions you can use to build a query (name == "Joe"), and groups can be used to group
predicates and more groups together. Groups can apply and or or operators to its nested
components.
# Add a predicate to the root of the form (which is itself a group)
filter_form = AshPhoenix.add_predicate(filter_form, :some_field, :eq, "Some Value")

# Add a group and another predicate to that group
{filter_form, group_id} = AshPhoenix.add_group(filter_form, operator: :or, return_id?: true)
filter_form = AshPhoenix.add_predicate(filter_form, :another, :eq, "Other", to: group_id)
validate/1 is used to merge the submitted form params into the filter form, and one of the
provided filter functions to apply the filter as a query, or generate an expression map,
depending on your requirements:
filter_form = AshPhoenix.validate(socket.assigns.filter_form, params)

# Generate a query and pass it to the Domain
query = AshPhoenix.FilterForm.filter!(MyApp.Payroll.Employee, filter_form)
filtered_employees = MyApp.Payroll.read!(query)

# Or use one of the other filter functions
AshPhoenix.FilterForm.to_filter_expression(filter_form)
AshPhoenix.FilterForm.to_filter_map(filter_form)

  
    
  
  LiveView Example


You can build a form and handle adding and removing nested groups and predicates with the following:
alias MyApp.Payroll.Employee

@impl true
def render(assigns) do
  ~H"""
  <.simple_form
    :let={filter_form}
    for={@filter_form}
    phx-change="filter_validate"
    phx-submit="filter_submit"
  >
    <.filter_form_component component={filter_form} />
    <:actions>
      <.button>Submit</.button>
    </:actions>
  </.simple_form>
  <.table id="employees" rows={@employees}>
    <:col :let={employee} label="Payroll ID"><%= employee.employee_id %></:col>
    <:col :let={employee} label="Name"><%= employee.name %></:col>
    <:col :let={employee} label="Position"><%= employee.position %></:col>
  </.table>
  """
end

attr :component, :map, required: true, doc: "Could be a FilterForm (group) or a Predicate"

defp filter_form_component(%{component: %{source: %AshPhoenix.FilterForm{}}} = assigns) do
  ~H"""
  <div class="border-gray-50 border-8 p-4 rounded-xl mt-4">
    <div class="flex flex-row justify-between">
      <div class="flex flex-row gap-2 items-center">Filter</div>
      <div class="flex flex-row gap-2 items-center">
        <.input type="select" field={@component[:operator]} options={["and", "or"]} />
        <.button phx-click="add_filter_group" phx-value-component-id={@component.source.id} type="button">
          Add Group
        </.button>
        <.button
          phx-click="add_filter_predicate"
          phx-value-component-id={@component.source.id}
          type="button"
        >
          Add Predicate
        </.button>
        <.button
          phx-click="remove_filter_component"
          phx-value-component-id={@component.source.id}
          type="button"
        >
          Remove Group
        </.button>
      </div>
    </div>
    <.inputs_for :let={component} field={@component[:components]}>
      <.filter_form_component component={component} />
    </.inputs_for>
  </div>
  """
end

defp filter_form_component(
       %{component: %{source: %AshPhoenix.FilterForm.Predicate{}}} = assigns
     ) do
  ~H"""
  <div class="flex flex-row gap-2 mt-4">
    <.input
      type="select"
      options={AshPhoenix.FilterForm.fields(Employee)}
      field={@component[:field]}
    />
    <.input
      type="select"
      options={AshPhoenix.FilterForm.predicates(Employee)}
      field={@component[:operator]}
    />
    <.input field={@component[:value]} />
    <.button
      phx-click="remove_filter_component"
      phx-value-component-id={@component.source.id}
      type="button"
    >
      Remove
    </.button>
  </div>
  """
end

@impl true
def mount(_params, _session, socket) do
  socket =
    socket
    |> assign(:filter_form, AshPhoenix.FilterForm.new(Employee))
    |> assign(:employees, Employee.read_all!())

  {:ok, socket}
end

@impl true
def handle_event("filter_validate", %{"filter" => params}, socket) do
  {:noreply,
   assign(socket,
     filter_form: AshPhoenix.FilterForm.validate(socket.assigns.filter_form, params)
   )}
end

def handle_event("filter_submit", %{"filter" => params}, socket) do
  filter_form = AshPhoenix.FilterForm.validate(socket.assigns.filter_form, params)

  case AshPhoenix.FilterForm.filter(Employee, filter_form) do
    {:ok, query} ->
      {:noreply,
       socket
       |> assign(:employees, Employee.read_all!(query: query))
       |> assign(:filter_form, filter_form)}

    {:error, filter_form} ->
      {:noreply, assign(socket, filter_form: filter_form)}
  end
end

def handle_event("remove_filter_component", %{"component-id" => component_id}, socket) do
  {:noreply,
   assign(socket,
     filter_form:
       AshPhoenix.FilterForm.remove_component(socket.assigns.filter_form, component_id)
   )}
end

def handle_event("add_filter_group", %{"component-id" => component_id}, socket) do
  {:noreply,
   assign(socket,
     filter_form: AshPhoenix.FilterForm.add_group(socket.assigns.filter_form, to: component_id)
   )}
end

def handle_event("add_filter_predicate", %{"component-id" => component_id}, socket) do
  {:noreply,
   assign(socket,
     filter_form:
       AshPhoenix.FilterForm.add_predicate(socket.assigns.filter_form, :name, :contains, nil,
         to: component_id
       )
   )}
end

AshPhoenix.FilterForm.Arguments 
    



      
Represents the arguments to a calculation being filtered on

AshPhoenix.FilterForm.Predicate 
    



      
Represents an individual predicate appearing in a filter form.
Predicates are grouped up in an AshPhoenix.FilterForm to create boolean
filter statements.

AshPhoenix.Form.InvalidPath exception
    



      
Raised when an invalid path is used to find, update or remove a form

AshPhoenix.Form.NoActionConfigured exception
    



      
Raised when a form action should happen but no action of the appropriate type has been configured

AshPhoenix.Form.NoDataLoaded exception
    



      
Raised when a data needed to be used but the required data was not loaded

AshPhoenix.Form.NoFormConfigured exception
    



      
Raised when attempting to refer to a form but no nested form with that name was configured.

AshPhoenix.Form.NoResourceConfigured exception
    



      
Raised when a form needed to be constructed but the resource for that form could not be determined

mix ash_phoenix.gen.html 
    



      
This task renders .ex and .heex templates and copies them to specified directories.

  
    
  
  Positional Arguments


	domain - The domain (e.g. "Shop").
	resource - The resource (e.g. "Product").


  
    
  
  Options


	--resource-plural - The plural resource name (e.g. "products")

mix ash_phoenix.gen.html MyApp.Shop MyApp.Shop.Product --plural-name products

mix ash_phoenix.gen.live 
    



      
Generates liveviews for a given domain and resource.
The domain and resource must already exist, this task does not define them.

  
    
  
  Positional Arguments


	domain - The domain (e.g. "Shop").
	resource - The resource (e.g. "Product").


  
    
  
  Options


	--resource-plural - The plural resource name (e.g. "products")

For example:
mix ash_phoenix.gen.live ExistingDomainName ExistingResourceName

